

IBM Power Week 2025

#pw2025

18 - 19 - 20 novembre 2025

IBM Innovation Studio Paris

S64 – RPG Battle Dev 2025 : corrections et remise des prix

19 novembre 16:00 - 17:00

Philippe Bourgeois
IBM France
pbourgeois@fr.ibm.com

Jérôme Clément
COMMON France
jerome.clement@ocsigroup.fr

Florian Gradot
COMMON France
florian.gradot@gaia.fr

IBM

common
FRANCE

Plan de la présentation

- 1. La RPG Battle Dev 2025
 - Introduction
 - Restitution de la correction des épreuves
 - Classement et remise des prix
- 2. Les évolutions récentes du RPG

IBM Power Week

18-19-20 novembre 2025

IBM
common
FRANCE

IBM

1. La RPG Battle Dev 2025



BE THE BEST, BE THE WINNER !

La RPG Battle Dev 2025

- 3ème édition
- Un concours de développement en langage RPG ouvert à tous
- Organisé par IBM France et COMMON France
- A titre individuel ou en équipe
- En ligne
- En 3 manches :
 - Du 08 au 10 octobre : épreuves E1 et E2
 - Du 15 au 17 octobre : épreuves E3 et E4
 - Du 05 au 07 novembre : épreuves E5 et E6
- Gratuit
- En français
- Sur une partition IBM i de PUB400

L'équipe organisatrice

Philippe BOURGEOIS



Jérôme CLÉMENT



Florian GRADOT



Quelques chiffres pour cette 3ème edition

- 43 participations
 - 16 équipes de 2 à 5 personnes
 - 27 participants individuels
 - **84** développeurs
- De 9 pays
 - Afrique du Sud
 - Belgique
 - Canada
 - France
 - Espagne
 - Italie
 - Maroc
 - Sénégal
 - Suisse

Prévoyance
Touftouf
ToutFaire
Bet'Adis
Alph'ADIS
Les d'jeuns
Altinate
Les Pépins du XVI
Hardis TSI
IBMi Guild
GO GLS
Equipe MBE
la i team
PralineSeries
Décodeurs Haute Loire
400 Team INFITEX

JohnLeFoudroyeur
Stanhomes France
FDZ
DASOW
Pab
AZ1
BENERGY
dev
Sathish Kumar
MLR
Eureka
Yo
RPGiste Castelroussin
YLH-TEAMS
Iliana Ros
azria alain
Clément GIRARD

Barpat
Cariboo
CHRISTIAN LARSEN VALVERDE
Georges Kopp
MK1
ZinzinDuRPG
8euros
Logikoum
JPMS
Romain Duloquin



d'avoir participé

IBM Power Week

18-19-20 novembre 2025



IBM

Présentation





Club des utilisateurs IBM i & Power

Objectifs

- Partager la compétence
- Favoriser l'échange d'informations
- Mutualiser la veille technologique
- Faire vivre la communauté IBM i
- ...

Avantages des membres

- Participer gratuitement aux événements
- Profiter des contenus exclusifs
- Accès à la base de connaissances Common
- Tarif sur le Congrès Annuel de Common Europe
- ...



IBM Power Week

18-19-20 novembre 2025



Temps forts 2025

- POWERLAB – Créer et traiter des données en JSON avec SQL
- Webinaire IBM i & Cloud
- Pause Café #94
- Challenge EBCDIC PLAGE
- RPG Battle Dev en partenariat avec IBM
- IBM Power Week en partenariat avec IBM
- ...



Restitution de la correction des épreuves



BE THE BEST, BE THE WINNER !

Nos critères de correction

- Programme rendu dans les temps ?
 - Bon résultat ?
 - Code simple et moderne ?
 - Code lisible ?
 - Code optimisé ?
-
- Nous étions 3 à nous partager les corrections, de façon anonyme, de vos 258 programmes ... avec le plus d'objectivité possible ... mais aussi le souci du détail pour vous départager

Manche 1 (8 octobre) – Epreuve E1 – Quelques nouveautés RPG

```
// 1. Afficher, par SND-MSG, la date (du travail) du jour au format AAAA/MM/JJ  
snd-msg %char(%date():*longjobrun); 2025/11/05
```

Date format	DATFMT	<u>*YMD</u>
Date separator	DATSEP	<u>'/'</u>

```
// 2. Afficher, par SND-MSG, l'horodate (système) du jour sous la forme AAAA-MM-JJ-HH.MN.SS  
snd-msg %char(%timestamp(*sys:0)); 2025-11-05-16.12.41
```

```
dcl-s langages varchar(10) dim(*auto:20);  
langages = %list('CL':'RPG':'COBOL':'C':'C++':'Java':'Python');  
  
// 3. Afficher, par SND-MSG, les valeurs du tableau LANGAGES sous la forme CL;RPG;COBOL;C;C++;Java;Python  
snd-msg %concatarr(';':langages); CL;RPG;COBOL;C;C++;Java;Python
```

Manche 1 (8 octobre) – Epreuve E1 – Quelques nouveautés RPG

```
ctl-opt decedit('0,');
```

```
dcl-s ts1 timestamp inz(z'2025-10-08-14.20.30.666');
```

```
dcl-s ts2 timestamp inz(z'2025-10-08-14.20.32.000');
```

// 4. Afficher, par SND-MSG, la différence exacte en secondes entre TS2 et TS1 : 1,334

```
snd-msg %char(%diff(ts2:ts1:*s:3)); 1,334
```

```
ctl-opt dclopt(*nochgdslen);
```

```
dcl-ds ds1;  
  z1 char(10);  
end-ds;
```

Si non spécifié

```
dcl-s z2 char(%size(ds1));
```

ID	Message
⊗ RNF3320	Le paramètre du mot clé n'est pas défini. Le mot clé n'est pas pris en compte.

// 5. Décommenter la déclaration de la zone Z2, puis afficher,
// par SND-MSG, la taille occupée en mémoire de la zone Z2 : 10

```
snd-msg %char(%size(z2)); 10
```

Manche 1 (8 octobre) – Epreuve E1 – Quelques nouveautés RPG

```
dcl-s z3  varchar(100) inz('"RPG Battle Dev 2025" - Organisé par COMMON France et IBM');  
dcl-s z3w like(z3);  
  
// 6. Afficher, par SND-MSG, le dernier mot du texte entre guillemets de la zone Z3 : 2025  
z3w = %left(z3 : %scanr('"' : z3)-1);  
snd-msg %subst(z3w : %scanr(' ' : z3w)+1); 2025
```

```
ctl-opt expropts(*usedecedit:*alwblanknum) decedit('0,');  
  
dcl-s z4  varchar(8);  
  
// 7. Afficher, par SND-MSG, le résultat de la conversion en décimal de la zone Z4,  
// dans les 2 cas suivants :  
z4 = '1.584,73';  
// Le SND-MSG doit renvoyer 1584,73  
snd-msg %char(%dec(z4:8:2)); 1584,73  
  
z4 = *blanks;  
// Le SND-MSG doit renvoyer 0,00  
snd-msg %char(%dec(z4:5:2)); 0,00
```

Manche 1 (8 octobre) – Epreuve E1 – Quelques nouveautés RPG





```
dcl-s z5  varchar(20) ccsid(*utf8) inz('áëiou');
```

```
// 8. Afficher, par SND-MSG, la longueur du contenu de la zone Z5 : 5  
//      puis afficher, par SND-MSG, les 3 premiers caractères de la zone Z5 : áëi  
snd-msg %char(%charcount(z5));  
snd-msg %left(z5:3:*natural);
```

```
ctl-opt bnddir('TEAMLIB/BATTLE2025');
```

```
dcl-pr entiersToRomains varchar(30) dim(5);  
|   tEntiers int(5) dim(5) value;  
end-pr;
```

```
dcl-s z6 varchar(30) dim(5);  
dcl-s romain varchar(30);
```

▼  TEAMLIB.*lib.prod-usr
▼  BATTLE2025.*srvpgm.rpgle
▼  BATTLE2025.*module
  ENTIERSTOROMAINS

COMFR.POWERBUNKER.COM:TEAMLIB/BATTLE2025 OBJTYPE(*BNDDIR)

Nom	Type	Bibliothèque	Activation
▼ BATTLE2025	*SRVPGM	TEAMLIB	*IMMED
ENTIERSTOROMAINS	*EXPORT		

Manche 1 (8 octobre) – Epreuve E1 – Quelques nouveautés RPG

```
// 9. Dans le programme de service TEAMLIB/BATTLE2025 vous disposez d'une procédure entiersToRomains
// qui prend une liste d'entiers en paramètre
// et qui renvoie la liste de ces entiers convertis en chiffres romains
// Voici le prototype de cette procédure :
//   dcl-pr entiersToRomains varchar(30) dim(5);
//       tEntiers int(5) dim(5) value;
//   end-pr;
// Ce programme de service est référencé dans le répertoire de liage TEAMLIB/BATTLE2025
//
// En utilisant cette procédure entiersToRomains, afficher par SND-MSG le résultat de la conversion
// des 5 entiers suivants : 18, 400, 944, 1237 et 3064
// Le résultat attendu est le suivant :
//   XVIII
//   CD
//   CMXLIV
//   MCCXXXVII
//   MMMLXIV
```

```
z6 = entiersToRomains(%list(18:400:944:1237:3064));
for-each romain in z6;
|   snd-msg romain;
endfor;

*inlr = *on;
```

```
XVIII
CD
CMXLIV
MCCXXXVII
MMMLXIV
```

Manche 1 (8 octobre) – Epreuve E2 – Vols au départ de New-York

■ Enoncé

```
// RPG BATTLE DEV 2025 - Manche 1 - Epreuve 2 - Du 8 au 10 octobre 2025

// Objectif : afficher la liste des vols possibles depuis un aéroport de départ

// Contexte :
// - La table TEAMLIB/VOLS contient la liste des vols (2 colonnes :
//   aéroport de départ (DEPART), aéroport d'arrivée (ARRIVEE))
// - Vous disposez également de l'index TEAMLIB/VOLSBYDEP
//   (clé sur la colonne DEPART en ascendant)
```

■ Résultat attendu

```
Vols au départ de : New York
New York > Paris
New York > Paris > Madrid
New York > Paris > Caire
New York > Paris > Rome
New York > Londres
New York > Londres > Athenes
New York > Londres > Athenes > Nicosie
New York > Los Angeles
New York > Los Angeles > Tokyo
New York > Los Angeles > Tokyo > Hawaii
```

Table VOLS

DEPART	ARRIVEE
New York	Paris
Chicago	Miami
New York	Londres
Londres	Athenes
Athenes	Nicosie
Paris	Madrid
Paris	Caire
Chicago	Francfort
Francfort	Moscou
Francfort	Pekin
Moscou	Tokyo
Francfort	Vienne
Paris	Rome
Miami	Lima
New York	Los Angeles
Los Angeles	Tokyo
Tokyo	Hawaii
Washington	Toronto
Caire	Paris

Manche 1 (8 octobre) – Epreuve E2 – Vols au départ de New-York

- Proposition de correction (1/2)

```
ctl-opt actgrp(*new);

dcl-f volsbydep rename(vols:f vols) keyed data(*nocvt);

dcl-s origine like(depart);

// Exemple 1
origine = 'New York';
snd-msg ('Vols au départ de : ' + origine);
affichageVol(origine:origine);
```

Manche 1 (8 octobre) – Epreuve E2 – Vols au départ de New-York

■ Proposition de correction (2/2)

```
// Procédure d'affichage des vols à partir de la ville d'origine
// La procédure est appelée de façon récursive

// On teste si on boucle (cycle),
// c'est à dire le cas suivant :
//   ville_depart1 --> ville_arrivée1 --> ville_depart1

dcl-proc affichageVol;

  dcl-pi *n;
  | ville   like(depart);
  | chemin  varchar(200) const;
  end-pi;

  dcl-ds dsVol likerec(fvols);
  dcl-ds tVols likeds(dsVol) dim(*auto:100);
  dcl-s nouveauChemin like(chemin);
  dcl-s indice int(5);
```

```
// Remplissage des différentes destinations à partir d'une ville d'origine
setll ville volsbydep;
if %equal;
  | reade ville volsbydep dsVol;
  | dow not %eof;
  |   | tVols(*next) = dsVol;
  |   | reade ville volsbydep dsVol;
  | enddo;
endif;

// Affichage des différents chemins
for indice=1 to %elem(tVols);
  | if %scan(tVols(indice).arrivee : chemin) = 0; // Pour éviter de boucler
  |   | nouveauChemin = %concat(' > ' : chemin : tVols(indice).arrivee);
  |   | snd-msg nouveauChemin;
  |   | affichageVol(tVols(indice).arrivee:nouveauChemin); // Appel récursif
  | endif;
endfor;
```

Manche 1 (8 octobre) – Epreuve E2 – *Vols au départ de New-York*

- Vérification avec d'autres valeurs

```
// Exemple 2
origine = 'Francfort';
snd-msg ('Vols au départ de : ' + origine);
affichageVol(origine:origine);

// Exemple 3
origine = 'Paris';
snd-msg ('Vols au départ de : ' + origine);
affichageVol(origine:origine);

*inlr = *on;
```

```
Vols au départ de : Francfort
Francfort > Moscou
Francfort > Moscou > Tokyo
Francfort > Moscou > Tokyo > Hawaii
Francfort > Pekin
Francfort > Vienne
```

```
Vols au départ de : Paris
Paris > Madrid
Paris > Caire
Paris > Rome
```

Manche 2 (15 octobre) – Epreuve E3 – Tour de France

■ Enoncé

```
// RPG BATTLE DEV 2025 - Manche 2 - Epreuve 1 - Du 15 au 17 octobre 2025

// Objectif : afficher le top 3 des coureurs cyclistes du Tour de France
// avec l'écart en minutes et secondes

// Contexte :
// - La table TEAMLIB/ETAPES contient la liste des temps des coureurs pour chaque étape du tour
//   3 colonnes : le numéro de l'étape (NUM_ETP), le numéro du dossard (DOSSARD)
//               et la durée de l'étape réalisée par le coureur (DUREE)
// - Si la colonne DUREE contient la valeur nulle c'est que le coureur a abandonné l'étape
```

■ Résultat attendu

```
OR : dossard numéro 2
ARGENT : dossard numéro 4 : + 8 mn 55 secondes
BRONZE : dossard numéro 6 : + 35 mn 31 secondes
```

Table ETAPES

NUM_ETP	DOSSARD	DUREE
1	1	02.13.45
1	2	02.15.03
1	3	02.36.16
1	4	02.05.54
1	5	*NULL
1	6	02.45.28
2	1	03.12.11
2	2	03.12.12
2	3	*NULL
2	4	03.25.32
2	5	03.24.45
2	6	03.20.26
3	1	*NULL
3	2	02.45.19
3	3	02.40.25
3	4	02.50.03
3	5	02.48.52
3	6	02.42.11

Manche 2 (15 octobre) – Epreuve E3 – Tour de France

■ Proposition de correction (1/3)

```
ctl-opt alwnull(*usrctl) actgrp(*new);

dcl-f etapes rename(etapes:fetapes);

dcl-ds tResultat dim(*auto:200) qualified inz;
  numDossard like(dossard); // Numéro du dossard
  cumulTemps uns(5);       // Temps cumulé de toutes les étapes
  abandon ind;             // Abandon = *ON si abandon
end-ds;

dcl-s vHeure time; // Initialisée à t'00.00.00'

// Lecture de toutes les étapes et ajout des temps cumulés
// dans le tableau tResultat
read etapes;
dow not %eof;
  ajoutEtape();
  read etapes;
enddo;

// Affichage du podium
affichagePodium();
```

```
// Ajout des temps cumulés dans le tableau tResultat
dcl-proc ajoutEtape;

  dcl-s indice uns(5);

  indice = %lookup(dossard:tResultat(*).numDossard);

  if indice = 0; // Si le numéro de dossard n'existe pas
    indice = %elem(tResultat) + 1;
    tResultat(indice).numDossard = dossard;
  endif;

  if %nullind(duree); // Si le coureur a abandonné
    tResultat(indice).abandon = *on;
  else;
    tResultat(indice).cumulTemps += %diff(duree:vHeure:*s);
  endif;
end-proc;
```

Manche 2 (15 octobre) – Epreuve E3 – *Tour de France*

- Proposition de correction (2/3)

```
// Affichage du résultat du podium (les 3 premiers)
dcl-proc affichagePodium;

    dcl-s medaille varchar(10);
    dcl-s message  varchar(100);
    dcl-s cpt      uns(5);

    sorta(a) tResultat %fields(abandon:cumulTemps);

    for-each medaille in %list('OR':'ARGENT':'BRONZE');
        cpt +=1;
        message = %concat('... : ' : medaille : 'dossard numéro ' + %char(tResultat(cpt).numDossard));
        if cpt > 1;
            message = %concat('... : ' : message :
                               ' + ' + affichageEcart(tResultat(cpt).cumulTemps - tResultat(1).cumulTemps));
        endif;
        snd-msg message;
    endfor;
end-proc;
```


Manche 2 (15 octobre) – Epreuve E3 – *Tour de France*

- Proposition de correction (3/3)

```
// Affichage du temps d'écart entre les coureurs en minutes et secondes
dcl-proc affichageEcart;
|
  dcl-pi *n varchar(50);
  t1 uns(5) const;
end-pi;

  reset vHeure;
  vHeure += %seconds(t1);
  return %concat(*none : %char(%subdt(vHeure:*mn)) : ' mn ' : %char(%subdt(vHeure:*s)) : ' secondes');
  // OU return %trim(%editw(%dec(vHeure) : '      &mn& &secondes'));
end-proc;
```

Manche 2 (15 octobre) – Epreuve E4 – *Poème de Victor Hugo*

■ Enoncé

```
// RPG BATTLE DEV 2025 - Manche 2 - Epreuve 2 - Du 15 au 17 octobre 2025

// Objectif : afficher le poème "Demain dès l'aube" de Victor Hugo

// Contexte :
// La table TEAMLIB/MEFTXTF00 contient le texte du poème sur une seule ligne
// 1 colonne : texte à mettre en forme (TEXTE_BRUT)

// Les règles de mise en forme sont les suivantes :
// - 2 espaces consécutifs : saut de ligne à formaliser par la chaîne '-----'
// - Majuscule : passage à la ligne suivante si la majuscule est précédée d'un point + espace ('. ')
//               ou d'une virgule + espace (', ')
// - Corriger la faute : != 'V'
```

TEXTE_BRUT

Demain, dès l'aube Demain, dès l'aube, à l'heure où blanchit la campagne, Je partirai. !ois-tu, je sais que tu m'attends. J'irai par la forêt, j'irai par la montagne.
Je ne puis demeurer loin de toi plus longtemps. Je marcherai les yeux fixés sur mes pensées, Sans rien voir au dehors, sans entendre aucun bruit, Seul, inconnu, le dos courbé,
les mains croisées, Triste, et le jour pour moi sera comme la nuit. Je ne regarderai ni l'or du soir qui tombe, Ni les voiles au loin descendant vers Harfleur, Et quand j'arriverai,
je mettrai sur ta tombe, Un bouquet de houx vert et de bruyère en fleur. Victor Hugo.

Manche 2 (15 octobre) – Epreuve E4 – *Poème de Victor Hugo*

- Résultat attendu

```
Demain, dès l'aube
-----
Demain, dès l'aube, à l'heure où blanchit la campagne,
Je partirai. Vois-tu, je sais que tu m'attends.
J'irai par la forêt, j'irai par la montagne.
Je ne puis demeurer loin de toi plus longtemps.
-----
Je marcherai les yeux fixés sur mes pensées,
Sans rien voir au dehors, sans entendre aucun bruit,
Seul, inconnu, le dos courbé, les mains croisées,
Triste, et le jour pour moi sera comme la nuit.
-----
Je ne regarderai ni l'or du soir qui tombe,
Ni les voiles au loin descendant vers Harfleur,
Et quand j'arriverai, je mettrai sur ta tombe,
Un bouquet de houx vert et de bruyère en fleur.
-----
Victor Hugo.
```

Manche 2 (15 octobre) – Epreuve E4 – *Poème de Victor Hugo*

- Proposition de correction (1/2)

```
**FREE
ctl-opt actgrp(*new) openopt(*nocvtdata);

dcl-f meftxtf00 extdesc('TEAMLIB/MEFTXTF00') extfile(*extdesc) alias;

dcl-s text_modifie      varchar(1000);
dcl-s ligne_texte       varchar(100);
dcl-s lettre            char(1);
dcl-s ponctuation       char(1);
dcl-s ligne_separation  char(20) inz(*all'-');
dcl-c separateur '|';

dcl-ds *n;
  alphabet      char(26) inz('ABCDEFGHIJKLMNOPQRSTUVWXYZ');
  alphabetLettre char(1)  dim(26) overlay(alphabet);
end-ds;
```

Manche 2 (15 octobre) – Epreuve E4 – *Poème de Victor Hugo*

- Proposition de correction (2/2)

```
// Lecture du texte à formater
read meftxtf00;
if %error or %eof(meftxtf00);
    return;
endif;
```

```
// Remplacement des doubles espaces par une ligne de tirets
text_modifie = %scanrpl(' '
                        :separateur + ligne_separation + separateur
                        :%trim(TEXTE_BRUT));

// Ajout des séparateurs de lignes
// pour chaque majuscule précédée de '.' ou ','
for-each lettre in alphabetLettre;
    for-each ponctuation in %list('.', ',');
        text_modifie = %scanrpl(ponctuation + ' ' + lettre
                                :ponctuation + separateur + lettre
                                :text_modifie);
    endfor;
endfor;

// Correction '!' à remplacer par 'V'
text_modifie = %scanrpl('!' : 'V' : text_modifie);

// Envoi en joblog chaque ligne de texte (découpage à chaque separateur)
for-each ligne_texte in %split(text_modifie : separateur);
    snd-msg ligne_texte;
endfor;

*inlr = *on;
```

Manche 3 (5 novembre) – Epreuve E5 – Statut d'un film

■ Enoncé E5A

```
// RPG BATTLE DEV 2025 - Manche 3 - Epreuve 1A - Du 5 au 7 novembre 2025
//
// Cette épreuve est composée de 2 parties (A et B), avec 2 membres source (E5A et E5B) :
// Partie A (source E5A) : création d'une procédure ILE, encapsulation dans un programme de service
// Partie B (source E5B) : appel de la procédure créée à la partie A
//
// Partie A - Source E5A

// Objectif : coder une procédure statutFilm qui prendra les caractéristiques suivantes :
//
// 1. Réception de 3 paramètres :
//   - TITRE de type VARCHAR(100)
//   - ETAT de type VARCHAR(10)
//   - DATESORTIE de type VARCHAR(20)
//
// 2. Lorsque l'on appellera la procédure dans la partie B et que l'on ne passera pas de valeur pour le paramètre DATESORTIE alors
//   la valeur 'inconnue' lui sera affecté
//
// 3. Lorsque l'on appellera la procédure dans la partie B, on pourra passer les codes suivants pour le paramètres ETAT :
//   - 'C00' qui signifie CONCEPTION
//   - 'P00' qui signifie PREPARATION
//   - 'T00' qui signifie TOURNAGE
//   - 'P10' qui signifie POSTPRODUCTION
//   - 'D00' qui signifie DIFFUSION
//
// 4. L'information "Etat du tournage" pourra prendre les valeurs suivantes :
//   - 'Tournage non démarré' si ETAT vaut C00 ou P00
//   - 'Tournage en cours' si ETAT vaut T00
//   - 'Tournage terminé' si ETAT vaut P10 ou D00
//   - 'Etat du tournage inconnu' si l'on passe la valeur nulle pour le paramètre ETAT
//
// 5. La procédure renverra la chaîne suivante : TITRE suivi de "Etat du tournage" suivi de DATESORTIE :
//   - Par exemple pour le film "DUNE 3", qui est actuellement en "TOURNAGE" et qui sortira le "16/12/2026", la procédure renverra
//     DUNE 3 - Tournage en cours - Date de sortie : 16/12/2026
//   - Autre exemple de sortie : STAR WARS: DAWN OF THE JEDI - Tournage non démarré - Date de sortie : inconnue
//   - Vous trouverez d'autres cas de figure dans l'énoncé de l'exercice E5B
//
// 6. La procédure devra être encapsulée dans un programme de service intitulé E5A et devra avoir la signature suivante : RPG_BATTLE_DEV_2025
```

Manche 3 (5 novembre) – Epreuve E5 – Statut d'un film

■ Enoncé E5B

```
// RPG BATTLE DEV 2025 - Manche 3 - Epreuve 1B - Du 5 au 7 novembre 2025
//
// Cette épreuve est composée de 2 parties (A et B), avec 2 membres source (E5A et E5B) :
//  Partie A (source E5A) : création d'une procédure ILE, encapsulation dans un programme de service
//  Partie B (source E5B) : appel de la procédure créée à la partie A
//
// Partie B - Source E5B

// Objectif : coder un programme E5B qui testera la procédure STATUTFILM créée dans la partie A
// Tester différents appels de la procédure en envoyant des messages dans la joblog par SND-MSG. Exemples :

// Avec les informations 'DUNE 3', 'T00' et '16/12/2026', le message dans la joblog devra être : DUNE 3 - Tournage en cours - Date de sortie : 16/12/2026

// Avec les informations 'Star Wars: Dawn of the Jedi' et 'C00'
// le message dans la joblog devra être : Star Wars: Dawn of the Jedi - Tournage non démarré - Date de sortie : inconnue

// Avec les informations 'AVATAR 4', 'P00' et la date en cours + 4 ans + 1 mois + 15 jours
// le message dans la joblog devra être : AVATAR 4 - Tournage non démarré - Date de sortie : 2029-12-20 (20, 21, ou 22 en fonction de la date à laquelle
// vous exécutez le programme)

// Avec les informations 'The Batman: Part II', "la valeur nulle" et '29 septembre 2027'
// le message dans la joblog devra être : The Batman: Part II - Etat du tournage inconnu - Date de sortie : 29 septembre 2027

// Avec les informations 'L'Odyssée', 'P10' et 'd'2026-07-15'
// le message dans la joblog devra être : L'Odyssée - Tournage terminé - Date de sortie : 2026-07-15
```

Manche 3 (5 novembre) – Epreuve E5 – *Statut d'un film*

- Résultat attendu

DUNE 3 - Tournage en cours - Date de sortie : 16/12/2026

Star Wars: Dawn of the Jedi - Tournage non démarré - Date de sortie : inconnue

AVATAR 4 - Tournage non démarré - Date de sortie : 2029-12-24

The Batman: Part II - Etat du tournage inconnu - Date de sortie : 29 septembre 2027

L'Odyssée - Tournage terminé - Date de sortie : 2026-07-15

Manche 3 (5 novembre) – Epreuve E5A – Statut d'un film

- Proposition de correction (1/4)

```
ctl-opt nomain alwnull(*usrctl);

dcl-proc statutFilm export;

  dcl-pi *n varchar(200);
    titre      varchar(100) options(*trim) const;           // Pour suppression des espaces
    etat       varchar(10)  options(*nullind) const;         // Pour passage de la valeur nulle
    dateSortie varchar(20)  options(*nopass:*convert) value; // Pour passage optionnel et transformation en alphanumérique
  end-pi;

  dcl-s etatTournage varchar(50);

  dcl-enum etatFilm;
    conception 'C00';
    preparation 'P00';
    tournage    'T00';
    postproduction 'P10';
    diffusion   'D00';
  end-enum;
```

Manche 3 (5 novembre) – Epreuve E5A – Statut d'un film

■ Proposition de correction (2/4)

```
if not %passed(dateSortie);
  dateSortie = 'inconnue';
endif;

select;
  when %nullind(etat);
    etatTournage = 'Etat du tournage inconnu';
  when %trim(etat) in %list(conception:preparation);
    etatTournage = 'Tournage non démarré';
  when %trim(etat) = tournage;
    etatTournage = 'Tournage en cours';
  when %trim(etat) in %list(postproduction:diffusion);
    etatTournage = 'Tournage terminé';
endsl;

return %concat(' - ' : titre : etatTournage : 'Date de sortie : ' + dateSortie);




end-proc;
```

```
// Autre solution :
if %nullind(etat);
  etatTournage = 'Etat du tournage inconnu';
endif;

select %trim(etat);
  when-in %list(conception:preparation);
    etatTournage = 'Tournage non démarré';
  when-is tournage;
    etatTournage = 'Tournage en cours';
  when-in %list(postproduction:diffusion);
    etatTournage = 'Tournage terminé';
endsl;
```

Manche 3 (5 novembre) – Epreuve E5A – Statut d'un film

- Proposition de correction (3/4)
 - a. Compilation du source E5A par CRTRPGMOD pour créer un objet ***MODULE**
 - b. Création d'un fichier source QSRVSRC par la commande CRTSRCPF
 - c. Ajout d'un source E5A de type **BND** dans ce fichier QSRVSRC pour indiquer la **procédure à exporter** et la **signature**



```
STRPGMEXP SIGNATURE('RPG_BATTLE_2025')
EXPORT     SYMBOL('STATUTFILM')
ENDPGMEXP
```

Manche 3 (5 novembre) – Epreuve E5A – Statut d'un film

- Proposition de correction (3/4)
 - d. Création du programme de service E5A

The screenshot illustrates the steps to create a service program in an IBM i environment. It shows a sequence of three windows:

- Top Window:** A project browser showing a folder named "E5A.*module" containing a subfolder "STATUTFILM". A "Créer" (Create) button is highlighted with a red box. Below it is a "Mettre à jour" (Update) button.
- Middle Window:** A menu with two options: "Programme ..." and "Programme de service...". The "Programme de service..." option is highlighted with a red box. A red arrow points from this option to the third window.
- Bottom Window:** A dialog box titled "Create Service Program (CRTSRVPGM)". It contains the following fields and values:

Service program:	SRVPGM	>	E5A
Library:		>	TEAMxx
Module:	MODULE	>	E5A
Library:		>	TEAMxx
Export:	EXPORT		*SRCFILE
Export source file:	SRCFILE		QSRVSRC
Library:			*LIBL
Export source member:	SRCMBR		*SRVPGM

A red arrow points from the "E5A.*srvpgm.rpgle" entry in the bottom-left project browser to the "Create Service Program" dialog box.

Manche 3 (5 novembre) – Epreuve E5A – Statut d'un film

- Proposition de correction (4/4)
 - e. Création du **répertoire de liage** BATTLE2025 par la commande CRTBNDDIR
 - f. Ajout du programme de service E5A dans ce répertoire de liage par la commande ADDBNDDIRE

The diagram illustrates the process of creating and populating a link library (BNDDIR) for the BATTLE2025 project. It consists of four main components connected by red arrows:

- BATTLE2025.*bnmdir**: The initial file to be created.
- Répertoire de liage - TEAMxx/BATTLE2025**: The link library being created. A button labeled **Ajouter une entrée (Invite)** is highlighted, leading to the next step.
- Navigation Tree**: A tree view showing the structure of the library. The path **TEAMxx.*lib.prod-cur** is expanded, showing **E5A.*srvpgm.rpgle** (highlighted), which contains **E5A.*module** and **STATUTFILM**.
- Répertoire de liage - TEAMxx/BATTLE2025**: A detailed view of the library contents. It shows a table with the following data:

Nom	Type	Bibliothèque	Activation..
✓ E5A	*SRVPGM	TEAMxx	*IMMED
STATUTFILM	*EXPORT		

Manche 3 (5 novembre) – Epreuve E5B – Statut d'un film

- Proposition de correction

```
ctl-opt actgrp(*new) bnmdir('BATTLE2025') alwnull(*usrctl);

dcl-pr statutFilm varchar(200);
  titre      varchar(100) options(*trim) const;          // Pour suppression des espaces
  etat       varchar(10)  options(*nullind) const;        // Pour passage de la valeur nulle
  dateSortie varchar(20)  options(*nopass:*convert) value; // Pour passage optionnel et transformation en alphanumérique
end-pr;

dcl-s etatFilm varchar(10) nullind;

etatFilm = 'T00';
snd-msg statutFilm('DUNE 3' : etatFilm : '16/12/2026');

etatFilm = 'C00';
snd-msg statutFilm('Star Wars: Dawn of the Jedi' : etatFilm);

etatFilm = 'P00';
snd-msg statutFilm(' AVATAR 4' : etatFilm : %date + %years(4) + %months(1) + %days(15));

%nullind(etatFilm) = *on; // On passe la valeur nulle
snd-msg statutFilm('The Batman: Part II' : etatFilm : '29 septembre 2027');
%nullind(etatFilm) = *off;

etatFilm = 'P10';
snd-msg statutFilm('L''Odyssée' : etatFilm : d'2026-07-15');

*inlr = *on;
```

Classement et remise des prix



BE THE BEST, BE THE WINNER !

Le classement 2025

- A l'issue de la manche numéro 1

A l'issue de la manche n°1

Rang	Participant / Equipe	Points /40
1	Cariboo	40
2	Les Pépins du XVI	39
2	8euros	39
2	Logikoum	39
5	Touftouf	38
5	ToutFaire	38
5	IBMi Guild	38
8	Bet'Adis	37
8	la i team	37
8	400 Team INFITEX	37
8	CHRISTIAN LARSEN VALVERDE	37

Le classement 2025

- A l'issue de la manche numéro 2

A l'issue de la manche n°2

Rang	Participant / Equipe	Points /80
1	ToutFaire	78
2	Les Pépins du XVI	77
3	IBMi Guild	76
3	400 Team INFITEX	76
5	Logikoum	74
6	la i team	71
6	BENERGY	71
6	CHRISTIAN LARSEN VALVERDE	71
6	8euros	71
10	Bet'Adis	70

Le classement 2025

https://commonfrance.fr/commonfr/wp/?page_id=1128

- A l'issue de la manche numéro 3
- Classement final

A l'issue de la manche n°3

Rang	Participant / Equipe	Points /120
1	ToutFaire	117.5
2	Les Pépins du XVI	115
3	400 team INFITEX	113
4	IBMi Guild	111.5
5	CHRISTIAN LARSEN VALVERDE	110
5	Logikoum	110
7	la i team	108
8	Bet'Adis	107.5
9	BENERGY	106
10	PralineSeries	105
10	Cariboo	105

Remise des prix



L'année prochaine ?

- Si nous devons organiser une 4ème edition de la RPG Battle Dev, que pourrions-nous améliorer ?
 - Communication OK
 - Accès à la partition IBM i de PUB400 OK
 - Nombre de manches (3) OK
 - Délai entre les manches (1 semaine) A augmenter : 2 semaines minimum
 - Nombre d'épreuves (6) OK
 - Délai pour réaliser les épreuves (3 jours) A augmenter : 5 jours (du me au di ?)
 - Difficulté des épreuves OK
 - Session de restitution (présentiel + distanciel) OK

2. Les évolutions récentes du langage RPG



Les évolutions RPG de novembre 2025

■ RPG Cafe

- <https://www.ibm.com/support/pages/node/1106229>

- Still to come later in 2025
 - [Define a variable like an enumeration](#)
- Fall 2025
 - [New date formats for conversions: *YYMD, *MDYY, and *DMYY](#)
 - [Implicit CCSID conversion for built-in functions](#)

Announcement

Note: An earlier announcement mentioned an enhancement that allows you to define a variable like an enumeration. That enhancement is not available yet. Watch this page for another announcement later in 2025. [Define a variable like an enumeration](#)

- **New Fall 2025: PTF enhancements for 7.6 and 7.5 :**
 - [New date formats for conversions: *YYMD, *MDYY, and *DMYY](#)
 - [Implicit CCSID conversion for built-in functions](#)
 - **A bonus enhancement without a separate page:** Diagnostic messages in the compiler will now have the full name of the field in error instead of being truncated with "...".

New:

RNF7030 30	25 002500	The name or indicator PRODUCT_VERSION_1_2 is not defined.
RNF7030 30	27 002700	The name or indicator PRODUCT_VERSION_1_4 is not defined.

Before:

RNF7030 30	25 002500	The name or indicator PRODUCT... is not defined.
RNF7030 30	27 002700	The name or indicator PRODUCT... is not defined.

7.5 PTFs for the Fall 2025 Enhancements:

- ILE RPG compiler: 5770WDS SJ08064

7.6 PTFs for the Fall 2025 Enhancements:

- ILE RPG compiler: 5770WDS SJ08065
- ILE RPG compiler, TGTRLS(V7R5M0): 5770WDS SJ08092

Améliorations RPG concernant les dates

- 3 nouveaux **formats** de date (*DMYY, *MDYY et *YYMD) pour les conversions
 - Année sur 4
 - Les mêmes séparateurs que *DMY, *MDY, *YMD, à savoir : / - . , &
 - Ne peuvent pas être utilisés pour la définition de variables de type date

PTFs for 7.5, and 7.6

7.5:

- ILE RPG compiler: 5770WDS SJ08064

7.6:

- ILE RPG compiler: 5770WDS SJ08065
- ILE RPG compiler, TGTRLS(V7R5M0): 5770WDS SJ08092

```
dcl-s datnum1 zoned(6) inz(251124);
dcl-s datnum2 zoned(8) inz(20251124);

snd-msg %char(%date : *dmy);
snd-msg %char(%date : *dmyy);

snd-msg %char(%date(datnum1:*ymd));
snd-msg %char(%date(datnum2:*yymd));

test(de) *yymd datnum2;
if not %error;
| snd-msg ('DATNUM2 est au format YYMD');
else;
| snd-msg ('DATNUM2 invalide');
endif;
```

```
24/11/25
24/11/2025
2025-11-24
2025-11-24
DATNUM2 est au format YYMD
```

Améliorations RPG concernant les CCSID

- Conversion **implicite** du **CCSID** dans les fonctions intégrées suivantes :
 - %CHECK, %CHECKR, %LOOKUP, %SCAN, %SCANR, %SCANRPL, %SPLIT, %TLOOKUP, %TRIM, %TRIML, %TRIMR, %XLATE

PTFs for 7.5, and 7.6

7.5:

- ILE RPG compiler: 5770WDS SJ08064

7.6:

- ILE RPG compiler: 5770WDS SJ08065
- ILE RPG compiler, TGTRLS(V7R5M0): 5770WDS SJ08092

```
**free
dcl-s z1  varchar(20) ccsid(*utf8) inz('áëiou');
dcl-s lettre char(1);

lettre = 'i';
snd-msg %char(%scan(lettre:z1:*natural));

*inlr = *on;
```

3

```
000600 snd-msg %char(%scan(lettre:z1:*natural));
```

Auparavant

ID	Message	
RNS9308	Compilation stopped. Severity 20 errors found in program.	50
RNF0353	The first and second parameters of %SCAN or %SCANR are not of the same type.	20

Améliorations RPG concernant les énumérations

Énumération typée : une énumération peut être définie avec un type de données

```
dcl-enum colorCodes int(3) qualified;  
  blue 1;  
  green 2;  
  yellow 3;  
  no_color -1 dft;  
end-enum;
```

Une des constantes de l'énumération peut être définie comme valeur par défaut

Disponibilité : fin 2025 ?

Une variable, un paramètre ou une valeur de retour peut être définie comme une énumération typée

```
dcl-s color like(colorCodes);  
dcl-s my_favourite_color like(colorCodes) inz(colorCodes.blue);
```

```
dcl-pr my_proc;  
  dcl-pi *n;  
    p1 like(colorCodes) const;  
    p2 like(colorCodes) const options(*exact);  
  end-pi;  
end-pr;
```

MERC

